# Maintainer: Felix Yan <felixonmars@archlinux.org>
# Contributor: Arch Haskell Team <arch-haskell@haskell.org>

_hkgname=aws
pkgname=haskell-aws
pkgver=0.24
pkgrel=87
pkgdesc="Amazon Web Services (AWS) for Haskell"
url="https://github.com/aristidb/aws"
license=("BSD")
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-aeson' 'haskell-attoparsec' 'haskell-base16-bytestring'
         'haskell-base64-bytestring' 'haskell-blaze-builder' 'haskell-byteable'
         'haskell-case-insensitive' 'haskell-cereal' 'haskell-conduit' 'haskell-conduit-extra'
         'haskell-cryptonite' 'haskell-data-default' 'haskell-http-conduit'
         'haskell-http-client-tls' 'haskell-http-types' 'haskell-lifted-base' 'haskell-memory'
         'haskell-monad-control' 'haskell-network' 'haskell-network-bsd' 'haskell-old-locale'
         'haskell-resourcet' 'haskell-safe' 'haskell-scientific' 'haskell-tagged'
         'haskell-unordered-containers' 'haskell-utf8-string' 'haskell-vector' 'haskell-xml-conduit')
makedepends=('ghc' 'haskell-quickcheck' 'haskell-errors' 'haskell-http-client'
             'haskell-http-client-tls' 'haskell-quickcheck-instances' 'haskell-tasty'
             'haskell-tasty-hunit' 'haskell-tasty-quickcheck' 'haskell-transformers-base')
source=("https://hackage.haskell.org/packages/archive/${_hkgname}/${pkgver}/${_hkgname}-${pkgver}.tar.gz")
sha512sums=('d453ea185a47b0fb93d2e8849ea81e1b0bae081aa81d280f94618d8fb357b6e4b71a43a6c8df0e4ba2b64390346daf50672ed9f9849cc6737a5ae58b5ac52af8')

build() {
    cd $_hkgname-$pkgver

    runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
        --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname --enable-tests \
        --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm \
            -f-examples
    runhaskell Setup build $MAKEFLAGS
    runhaskell Setup register --gen-script
    runhaskell Setup unregister --gen-script
    sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
    sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
    cd $_hkgname-$pkgver
    runhaskell Setup test || echo "Requires AWS credentials to test"
}

package() {
    cd $_hkgname-$pkgver

    install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
    install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
    runhaskell Setup copy --destdir="$pkgdir"
    install -D -m644 LICENSE "$pkgdir"/usr/share/licenses/$pkgname/LICENSE
    rm -f "$pkgdir"/usr/share/doc/$pkgname/LICENSE
}
